

Cavity QED Systems: Spectral Properties




Figure 1: Originally, a cavity-QED system was a system of atoms interacting with quantized light in the cavity.

Two-level systems (atoms) are described by **Qubits**, denoted by symbol S .

```
In[*]:= Let[Qubit, S]
```

```
In[*]:= Dimension[S]
```

```
Out[*]=
```

2

Bosonic modes (cavity photons) are described by **Bosons**, denoted by symbol c .

```
In[*]:= Let[Boson, c]
```

```
In[*]:= {Bottom[c], Top[c], Dimension[c]}
```

```
Out[*]=
```

{0, 5, 6}

Jaynes-Cummings Hamiltonian

The qubit has transition frequency Ω .

```
In[*]:= Let[Real, Ω]
Hq = (Ω / 2) S[3]
```

```
Out[*]=
```

$$\frac{\Omega}{2} S^z$$

The photon has frequency ω .

```
In[*]:= Let[Real, ω]
Hc = ω * Dagger[c] ** c
```

```
Out[*]=
```

$$\omega c^\dagger c$$

The qubit-photon coupling in the rotating-wave approximation (RWA).

```
In[*]:= Let[Real, g]
Hg = g * (S[4] ** c + Dagger[c] ** S[5])
```

```
Out[*]=
```

$$g (c S^+ + c^\dagger S^-)$$

Construct the total Hamiltonian.

```
In[*]:= HH = Hq + Hc + Hg
```

```
Out[*]=
```

$$\omega c^\dagger c + g (c S^+ + c^\dagger S^-) + \frac{\Omega}{2} S^z$$

Basic Method

Examine the matrix representation of the Hamiltonian.

```
In[*]:= mat = Matrix[HH, {c, S}];
mat // MatrixForm
```

```
Out[*] // MatrixForm =
```

$$\begin{pmatrix} \frac{\Omega}{2} & 0 & 0 & g & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & -\frac{\Omega}{2} & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & \omega + \frac{\Omega}{2} & 0 & 0 & \sqrt{2} g & 0 & 0 & 0 & 0 & 0 & 0 \\ g & 0 & 0 & \omega - \frac{\Omega}{2} & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 2\omega + \frac{\Omega}{2} & 0 & 0 & \sqrt{3} g & 0 & 0 & 0 & 0 \\ 0 & 0 & \sqrt{2} g & 0 & 0 & 2\omega - \frac{\Omega}{2} & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 3\omega + \frac{\Omega}{2} & 0 & 0 & 2g & 0 & 0 \\ 0 & 0 & 0 & 0 & \sqrt{3} g & 0 & 0 & 3\omega - \frac{\Omega}{2} & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 4\omega + \frac{\Omega}{2} & 0 & 0 & \sqrt{5} g \\ 0 & 0 & 0 & 0 & 0 & 0 & 2g & 0 & 0 & 4\omega - \frac{\Omega}{2} & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 5\omega + \frac{\Omega}{2} & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \sqrt{5} g & 0 & 0 & 5\omega - \frac{\Omega}{2} \end{pmatrix}$$

Write a small utility function to use to investigate the spectrum of the Hamiltonian.

```

In[*]:= spectrum[gv_] := Block[
  { $\omega$  = 1,
    $\Omega$  = 1.05,
   g},
  g = gv;
  val = Sort@Eigenvalues[Normal@mat]
]


```

Now, examine the level structure of the Hamiltonian and its evolution as coupling g varies.

```

In[*]:= LevelsPlot[{spectrum[0.], spectrum[0.1], spectrum[0.3], spectrum[0.4]},
  ImageSize -> Medium]

```




Note that there seems to be crossing of energy levels. Let us examine the spectrum more closely by varying g continuously.

```

In[*]:= data = Transpose@Table[spectrum[x], {x, 0, 0.5, 0.01}];
ListPlot[data,
  FrameLabel → {"g / ω", "Energy"},
  Joined → True, DataRange → {0, 0.4}, PlotStyle → Blue]

```

Out[*]=



Parity Conservation

The numerical results above suggest that energy levels cross each other. This is unusual unless there is a symmetry.

Question: Is there a symmetry? What is it if any?

Indeed, the parity is conserved.

```

In[*]:= op = Parity[{c, S}]

```

Out[*]=

Parity[c] S^z

```

In[*]:= op ** HH - HH ** op

```

Out[*]=

0

Then, let us take a look at the basis again.

```

In[*]:= bs = Basis[{c, S}]

```

Out[*]=

$\{ |0_c 0_s\rangle, |0_c 1_s\rangle, |1_c 0_s\rangle, |1_c 1_s\rangle, |2_c 0_s\rangle, \\ |2_c 1_s\rangle, |3_c 0_s\rangle, |3_c 1_s\rangle, |4_c 0_s\rangle, |4_c 1_s\rangle, |5_c 0_s\rangle, |5_c 1_s\rangle \}$

Note that each basis state is an eigenstate of the parity operator, each with different parities.

```
In[*]:= op ** bs
Out[*]= { |0c0s>, -|0c1s>, -|1c0s>, |1c1s>, |2c0s>,
          -|2c1s>, -|3c0s>, |3c1s>, |4c0s>, -|4c1s>, -|5c0s>, |5c1s> }
```

Let us group the basis elements according to their parities.

```
In[*]:= ps = GroupBy[bs, ParityValue[{c, S}]]
Out[*]= { 1 -> { |0c0s>, |1c1s>, |2c0s>, |3c1s>, |4c0s>, |5c1s> },
          -1 -> { |0c1s>, |1c0s>, |2c1s>, |3c0s>, |4c1s>, |5c0s> } }
```

Rearrange the basis so that basis states of the same parity come together, and then examine again the matrix.

```
In[*]:= bb = Catenate[ps];
new = MatrixIn[HH, bb];
new // MatrixForm
Out[*]//MatrixForm=
```

$$\begin{pmatrix} \frac{\Omega}{2} & g & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ g & \omega - \frac{\Omega}{2} & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & \frac{1}{2}(4\omega + \Omega) & \sqrt{3}g & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & \sqrt{3}g & 3\omega - \frac{\Omega}{2} & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & \frac{1}{2}(8\omega + \Omega) & \sqrt{5}g & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & \sqrt{5}g & 5\omega - \frac{\Omega}{2} & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & -\frac{\Omega}{2} & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & \omega + \frac{\Omega}{2} & \sqrt{2}g & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & \sqrt{2}g & 2\omega - \frac{\Omega}{2} & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \frac{1}{2}(6\omega + \Omega) & 2g & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 2g & 4\omega - \frac{\Omega}{2} & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 5\omega + \end{pmatrix}$$

The resulting matrix is indeed block diagonal!

Therefore, it is more efficient to handle each block separately.

```
In[*]:= mm = MatrixIn[HH, ps];
MatrixForm /@ mm
```

```
Out[*]=
```

$$\langle 1 \rightarrow \begin{pmatrix} \frac{\Omega}{2} & g & 0 & 0 & 0 & 0 \\ g & \omega - \frac{\Omega}{2} & 0 & 0 & 0 & 0 \\ 0 & 0 & \frac{1}{2} (4\omega + \Omega) & \sqrt{3} g & 0 & 0 \\ 0 & 0 & \sqrt{3} g & 3\omega - \frac{\Omega}{2} & 0 & 0 \\ 0 & 0 & 0 & 0 & \frac{1}{2} (8\omega + \Omega) & \sqrt{5} g \\ 0 & 0 & 0 & 0 & \sqrt{5} g & 5\omega - \frac{\Omega}{2} \end{pmatrix},$$

$$-1 \rightarrow \begin{pmatrix} -\frac{\Omega}{2} & 0 & 0 & 0 & 0 & 0 \\ 0 & \omega + \frac{\Omega}{2} & \sqrt{2} g & 0 & 0 & 0 \\ 0 & \sqrt{2} g & 2\omega - \frac{\Omega}{2} & 0 & 0 & 0 \\ 0 & 0 & 0 & \frac{1}{2} (6\omega + \Omega) & 2g & 0 \\ 0 & 0 & 0 & 2g & 4\omega - \frac{\Omega}{2} & 0 \\ 0 & 0 & 0 & 0 & 0 & 5\omega + \frac{\Omega}{2} \end{pmatrix} \rangle$$

Summary

Functions

- LevelsPlot
 - Sort
 - Parity, ParityValue
 - MatrixIn
-
- Qubit, Dimension
 - Boson, Bottom, Top, Dimension
 - Matrix, ExpressionFor
 - ProperValues, ProperStates, ProperSystem

Related Links

- Mahn-Soo Choi, Advanced Quantum Technologies 12, 2000085 (2020), “Exotic Quantum States of Circuit Quantum Electrodynamics in the Ultra-Strong Coupling Regime.”
- C. Dongni, S. Luo, Y.-D. Wang, S. Chesi, and Mahn-Soo Choi, Physical Review A 105, 022627 (2022), “Geometric manipulation of a decoherence-free subspace in atomic ensembles.”